
Tema forrige uke

I Lister som datastruktur
I quote
I Rekursjon på lister
I Høyereordens prosedyrer

I Prosedyrer som parameter
I Prosedyrer som returverdi

I lambda

2



Høyereordens sekvensoperasjoner

? (define (map proc items)
(if (null? items)

'()
(cons (proc (car items))

(map proc (cdr items)))))

? (map square '(-1 -2 0 4 -42))
→ (1 4 0 16 1764)

? (define (filter pred items)
(cond ((null? items) '())

((pred (car items))
(cons (car items)

(filter pred (cdr items))))
(else (filter pred (cdr items)))))

? (filter (lambda (x)
(and (> x 2) (< x 8)))

'(0 1 2 3 4 5 6 7 8 9))
→ (3 4 5 6 7)

Høyereordens sekvensoperasjoner

? (define (reduce proc init items)
(if (null? items)

init
(proc (car items)

(reduce proc init (cdr items)))))

? (reduce + 0 '(1 2 3 4 5)) → 15

? (define (map proc items) ;; map definert med reduce
(reduce (lambda (item rest)

(cons (proc item) rest))
'() items))

? (define (filter pred items) ;; filter definert med reduce
(reduce (lambda (item rest)

(if (pred item)
(cons item rest)
rest))

'() items))

Tema denne uka

I lambda, let og lokale variabler
I Dataabstraksjon
I Trær og hierarkiske strukturer:
I Lister av lister
I Rekursjon på trær

Lokale variabler og lambda

I Nå skal vi se på hvordan vi kan bruke
prosedyreapplikasjon med lambda for å få
lokale variabler.

Prosedyrer og parametre

? (lambda (x)
(if (even? x) "even!" "odd!"))

→ #<procedure>

? ((lambda (x)
(if (even? x) "even!" "odd!"))

7)
→ "odd!"

I lambda lager prosedyrer,

I kan være anonyme eller ikke.

I Parametrene til en prosedyre
(som x her) er en lokal variabel:

I navngitt variabel der bindingens
rekkevidde er prosedyrekroppen.

I Noen ganger ønsker vi å kunne
introdusere flere lokale variabler
internt i prosedyren: let.

Lokale variabler, let og lambda

Et enkelt eksempel

? (let ((a 2)
(b 3))

(* a b))
→ 6

;; syntaktisk sukker for:

? ((lambda (a b)
(* a b))

2 3)
→ 6

I let-uttrykk med n variabler.

I Ekvivalent med å kalle et lambda-
uttrykk med n parametre:

I Variabelbindingene gitt ved
prosedyreargumentene.

I Kroppene i uttrykkene ellers like.

I Navnet gjenspeiler matematisk
språkbruk (‘Let a be a vector.’).

Lokale variabler, let og lambda (forts.)

Et mer realistisk eksempel

? (percentages '(10 90 50 50)) → (5 45 25 25)

(define (percentages items)
(map (lambda (x) (/ x (/ (reduce + 0 items) 100)))

items))

(define (percentages items)
(let ((sum (/ (reduce + 0 items) 100)))

(map (lambda (x) (/ x sum))
items)))

(define (percentages items)
((lambda (sum)

(map (lambda (x) (/ x sum))
items))

(/ (reduce + 0 items) 100)))

Lokale variabler, let og lambda (forts.)

Generell form for let

(let ((〈var1〉 〈exp1〉)
(〈var2〉 〈exp2〉)
...

(〈varn〉 〈expn〉))
〈body〉)

Ekvivalent lambda-uttrykk

((lambda (〈var1〉 〈var2〉 ... 〈varn〉)
〈body〉)

〈exp1〉 〈exp2〉 ... 〈expn〉)

I Rekkevidden til variablene er kroppen til let-uttrykket.
I Verdiene (〈exp1〉 . . . 〈expn〉) beregnes ‘utenfor’ let-uttrykket:
I Variablene har ikke tilgang til hverandre under bindingen.

Vi øver oss på å forstå bindinger

? (define foo 42)

? (let ((x foo)
(y 1))

(list x y))

→ (42 1)

? (let ((x foo)
(y foo))

(list x y))
→ (42 42)

? (let ((x foo)
(y x))

(list x y))
→ error: x undefined

? (let ((foo 7)
(y foo))

(list foo y))
→ (7 42)

? (let ((foo 7))
(let ((y foo))

(list foo y)))
→ (7 7)

? (let* ((foo 7)
(y foo))

(list foo y))
→ (7 7)

Lokale variabler, lambda og let*

Generell form for let*

(let* ((〈var1〉 〈exp1〉)
(〈var2〉 〈exp2〉)
...

(〈varn〉 〈expn〉))
〈body〉)

Ekvivalent lambda-uttrykk

((lambda (〈var1〉)
((lambda (〈var2〉)

((lambda (〈varn〉)
〈body〉)

〈expn〉))
〈exp2〉))

〈exp1〉)

I Mens let binder variablene parallelt så gjør let* det sekvensielt.
I let* er en kortform for flere omsluttende let- eller lambda-uttrykk.

Dataabstraksjon

I Mye av fokuset så langt har vært på prosedyrer.

I Modularisering og abstraksjon av prosesser (beregninger).

I Like viktig er modularisering og abstraksjon av data (‘kunnskap’).

I Fremover skal vi bruke en del tid på å snakke om data, bl.a:
I Hvordan definere abstrakte og sammensatte datatyper.
I Abstraksjonsbarrierer.
I Hva er egentlig data?
I Prosedyrer som datastrukturer.
I Strukturerte data og hierarkiske data (trær)

Noen datatyper

Hvilke typer data kjenner vi i Scheme så langt?

Type Eksempel

Tall integer, real, rational 42, 3.1415, 2/3
Sekvens av tegn string "foo bar"
Symbol symbol 'foo, 'hello
Sannhetsverdi boolean #t, #f
Prosedyrer procedure +, (lambda (x) (* x x))
Par (2-tuple) pair (47 . 11)
Tom liste null '()

I I tillegg har vi lister: eksempel på en kompleks / sammensatt datatype,
bygget på par og den tomme lista.

I Fremover skal vi jobbe med å definere egne datatyper.

I Viktig skille mellom abstrakte datatyper og deres implementasjon.

Hva er data?

I Egentlig bare gitt ved grensesnittet sitt: konstruktor og selektorer.

I Datatypen manipuleres gjennom et abstrakt grensesnitt som skjuler
implentasjonsdetaljene: abstraksjonsbarrieren.

I Eksempel: Hvordan har vi forholdt oss til datatypen par så langt?

I Kun via grensesnittet den er definert med: cons, car og cdr.

I Denne dataabstraksjonen skjuler en datatypes interne representasjon.

I Abstrakt datatype vs konkret datarepresentasjon:

I Nøyaktig hvordan par er representert internt er ikke viktig, så lenge
grensesnittet oppfyller ‘kontrakten’ sin.

I Vi skal implementere en egen alternativ versjon av datatypen par.

Par som prosedyre!

(define (cons x y)
(lambda (message)

(cond ((= message 0) x)
((= message 1) y))))

(define (car proc)
(proc 0))

(define (cdr proc)
(proc 1))

? (cons 1 2)
→ #<procedure>

? (cdr (cons 1 2))
→ 2

? (car (cons 1 (cons 2 3)))
→ 1

I Selv datastrukturer kan
implementeres som prosedyrer!

I Par som prosedyre: Returnerer enten
car- eller cdr-verdien, avhengig av
hvilken ‘kode’ den får (0/1).

I Eksempel på såkalt message passing.

I car og cdr kaller par-prosedyren
(laget med cons) med ‘beskjed’ om
hvilken verdi som ønskes.

Par som prosedyre! (forts.)

(define (cons x y)
(lambda (message)

(cond ((= message 0) x)
((= message 1) y))))

(define (car proc)
(proc 0))

(define (cdr proc)
(proc 1))

? (cons 1 2)
→ #<procedure>

? (cdr (cons 1 2))
→ 2

? (car (cons 1 (cons 2 3)))
→ 1

I Prosedyren som cons returnerer
brukes bare som en container for å
huske parameterbindingene:

I car- og cdr-verdiene (x og y)
‘gjemmes’ via såkalt innkapsling:

I Closures i vanlig Lisp-sjargong, men
ikke i SICP.

I Via prosedyren som returneres av
cons får vi tilgang til variablene
utenfor det som ellers ville være
rekkevidden av dens binding
(prosedyrekroppen til cons)!

I Eksempel på konseptuelt interessant
teknikk vi skal jobbe mye med senere!

18



Par som prosedyre: versjon 2.

(define (cons x y)
(lambda (z) (z x y)))

(define (car p)
...)

(define (cdr p)
...)

? (define foo (cons 42 '()))

? (car foo)
→ 1

? (cdr foo)
→ '()

I Å skrive ferdig car og cdr her blir del
av oblig 2a.

I Prosedyrerepresentasjonene våre er
mindre effektive enn vanlige par.

I Men gode eksempler på hvor mye
man kan gjøre med bare prosedyrer.

I Senere i kurset skal vi se flere
praktiske eksempler på prosedyrer
som datastrukturer!

Lister av lister
I Closure property i SICP: Liste-elementer kan selv være lister . . .
I Som igjen kan bestå av nye lister.

? (list (list 1 2) 3 4)
→ ((1 2) 3 4)

Trær

I Lister har så langt representert
sekvenser.

I Lister av lister kan sees som
trær.

I Nyttig for å representere
hierarkisk strukturerte data.

21



Litt terminologi for trær

I Et tre er en type graf, en mengde
noder forbundet med kanter.

I Den øverste noden kalles rot.

I Interne noder har barn / døtre:
nye subtrær.

I Løvnoder har ikke barn (såkalt
terminalnoder).

I Noder har kun én forelder.

I Kantene kan ikke være sykliske.

I Binærtrær: nodene har maks to
døtre, venstre/høyre.

Lister som trær

I Flate lister: sekvenser.
I Lister av lister: kan sees som trær.

I Hvert element i en liste er en gren.
I Elementer som selv er lister er subtrær.
I Løvnodene i treet er de atomære

elementene som ikke er lister.

I PS: Har verdier kun på løvnodene her;
skal etterhvert lage en mer generell
implementasjon av trær.

Rekursjon på trær (eksempel 1:3)

I Eksempel: telle løvnoder (parallelt til length på sekvenser).

? (count-leaves '((1 2) 3 4)) → 4

I Må passe på at rekursjonen går ned i hver (element)liste.
I Må tenke på tre forskjellige situasjoner. Argumentet kan være:

I den tomme lista
I en cons-celle (et tre)
I noe annet (løvnode)

(define (count-leaves tree)
(cond ((null? tree) 0)

((pair? tree)
(+ (count-leaves (car tree))

(count-leaves (cdr tree))))
(else 1)))

Rekursjon på trær (eksempel 2:3)

? (fringe '((1 2) ((3) 4)))
→ (1 2 3 4)

(define (fringe tree)
(cond ((null? tree) '())

((pair? tree)
(append (fringe (car tree))

(fringe (cdr tree))))
(else (list tree))))

;; append kombinerer to lister;
;; sammenlikning med cons:

? (append '(1 2) '(3)) → (1 2 3)

? (cons '(1 2) '(3)) →

((1 2) 3)

I Et annet eksempel: samle opp
alle løvnodene i en flat liste.

I Typisk eksempel på rekursjon på
lister av lister:

I Vi ønsker å gjøre noe for hvert
atomære element (løvnodene).

I Må passe på at rekursjonen går
ned i hver liste.

I To basistilfeller: tomt tre eller
løvnode.

Rekursjon på trær (eksempel 3:3)

I map er definert for lister: tree-map kan defineres for lister av lister:

? (tree-map square '((1 2) 3 4)) → ((1 4) 9 16)

(define (tree-map proc tree)
(cond ((null? tree) '())

((pair? tree)
(cons (tree-map proc (car tree))

(tree-map proc (cdr tree))))
(else (proc tree))))

I Alternativt kan map brukes, i kombinasjon med rekursjon:

(define (tree-map proc tree)
(map (lambda (subtree)

(if (pair? subtree) (tree-map proc subtree) (proc subtree)))
tree))

I Hvorfor trengs det bare én rekursiv oppkalling i denne varianten?

Trerekursjon

I count-leaves, fringe, osv. vil gi en trerekursiv prosess.
I Fikk dårlig rykte da vi så på fib tidligere:

eksponentiell vekst og masse redundante beregninger.
I Prosesstreet gjenspeilet trinnene i en naiv algoritme.

I Her gjenspeiler det bare strukturen til inputdata.
I Kompleksiteten lineær i antall noder i input. Ingen redundans.

Veien videre

I Så langt: Kan behandle nøstede lister som trær,

I men bruker bare pair?, null?, cons, car og cdr direkte.

I Ingen abstraksjon!

I Vi skal definere abstrakte datatyper for å representere trær.

I Ikke bare løvnodene, men også de interne nodene kan ha verdier.

I Skal også jobbe med eksempler der rekursjonen velger å kun følge én
gren (i stedet for å gå ned i alle subtrær).

I Som praktiske eksempler skal vi se på hvordan både mengder og
komprimeringskoder kan implementeres effektivt som trær.

Neste uke(r)

I Mer om både dataabstraksjon og hierarkiske strukturer
I Praktisk eksempel 1: representasjon av mengder.
I Praktisk eksempel 2: Huffmankoding (hovedtema for oblig 2a).

